

DÃY SỐ - CẤP SỐ

I. Dãy số.

1. Cách cho một dãy số tự nhiên liên tiếp từ k đến m.

Cú pháp: `> $k..m;`

Ví dụ 1:

Cho dãy các số tự nhiên trong khoảng $[2; 9]$:

+ Dùng hàm '\$' trong Maple như sau:

`> $ 2..9;`

2, 3, 4, 5, 6, 7, 8, 9

Ví dụ 2:

Cho dãy gồm 6 số tự nhiên lẻ đầu tiên:

`> 2*i+1 $i=0..5;`

1, 3, 5, 7, 9, 11

2. Cho một dãy số bất kì dựa vào số hạng tổng quát

(dãy số cho bởi công thức).

• Cú pháp 1: `> seq(fomul(i), i=k..m);`

Trong đó: - fomul(i) là một công thức(biểu thức) theo i;
- k, m: là các số tự nhiên (hoặc số nguyên).

Ví dụ 1:

Tìm 5 số hạng đầu của mỗi dãy số sau:

a) Dãy số (u_n) với $u_n = \frac{2n^2 - 3}{n}$;

b) Dãy số (u_n) với $u_n = \sin^2 \frac{n\pi}{4} + \cos \frac{2n\pi}{3}$;

c) Dãy số (u_n) với $u_n = (-1)^n \cdot \sqrt{4^n}$.

(Bài tập 9/tr105 SGK ĐS> 11, nâng cao)

+ Xác định 5 số hạng đầu của dãy số ở câu a):

`> seq((2*i^2-3)/i, i=1..5);`

$-1, \frac{5}{2}, 5, \frac{29}{4}, \frac{47}{5}$

+ Xác định 5 số hạng đầu của dãy số ở câu b):

`> seq(sin(i*Pi/4)^2+cos(2*i*Pi/3), i=1..5);`

$0, \frac{1}{2}, \frac{3}{2}, \frac{-1}{2}, 0$

+ Xác định 5 số hạng đầu của dãy số ở câu c):

`> seq((-1)^i*sqrt(4^i), i=1..5);`

-2, 4, -8, 16, -32

Cũng có thể xác định công thức của số hạng tổng quát trước (bằng cách cho hàm số) rồi tìm các số hạng của dãy theo các bước sau:

- Xác định hàm theo công thức của số hạng tổng quát:

> **u:=i->(-1)^i*sqrt(4^i);**

$$u := i \rightarrow (-1)^i \sqrt{4^i}$$

- Tính 5 số hạng đầu của dãy :

> **seq(u(i), i=1..5);**

$$-2, 4, -8, 16, -32$$

- Có thể cho hiển thị tường minh như sau:

> **seq(u[i]=u(i), i=1..5);**

$$u_1 = -2, u_2 = 4, u_3 = -8, u_4 = 16, u_5 = -32$$

✎ Giải thích:

- Ta dùng u[i] để hiển thị số hạng u_i ; còn u(i) là giá trị của hàm số u tại i (bởi vì trước đó ta đã định nghĩa u là hàm số).

• **Cú pháp 2:** > **seq(fomul(i), i=k..m, n);**

Trong đó: - fomul(i) là một công thức(biểu thức) theo i;
- k, m: là các số tự nhiên (hoặc số nguyên dương);
- n: là bước nhảy(step) của i.

Ví dụ 2:

Tìm các số hạng u_1, u_3, u_5, u_7, u_9 của dãy số (u_n) với $u_n = \frac{2}{3n-2}$.

+ Ta cho hiển thị cụ thể các số hạng và kí hiệu theo cách đã làm trên:

Xác định hàm số u:

> **u:=i->2/(3*i-2);**

$$u := i \rightarrow \frac{2}{3i-2}$$

> **seq(u[i]=u(i), i=1..9, 2);**

$$u_1 = 2, u_3 = \frac{2}{7}, u_5 = \frac{2}{13}, u_7 = \frac{2}{19}, u_9 = \frac{2}{25}$$

Bước nhảy

• **Cú pháp 3:** > **seq(fomul(i), i=[X]);**

> **seq(fomul(i), i in [X]);**

Trong đó: - fomul(i) là một công thức(biểu thức) theo i;
- [X]: là danh sách hoặc tập hợp.

Ví dụ 3:

Cho dãy số (u_n) với $u_n = \frac{n}{n^2+1}$. Tính các số hạng $u_1; u_4; u_7; u_{10}; u_{13}$?

+ Đầu tiên ta xác định tập hợp X:

> **x:={3*i+1 \$i=0..4};**

$$X := \{1, 4, 7, 10, 13\}$$

+ Tính các số hạng của dãy:

> **seq(u[i]=i/(i^2+1), i in X);**

$$u_1 = \frac{1}{2}, u_4 = \frac{4}{17}, u_7 = \frac{7}{50}, u_{10} = \frac{10}{101}, u_{13} = \frac{13}{170}$$

3. Dãy số cho bởi hệ thức truy hồi.

Ví dụ: Cho dãy số (u_n) xác định bởi $u_1 = 1, u_2 = -2$ và $u_n = u_{n-1} - 2u_{n-2}$ với mọi $n \geq 3$. Tính $u_3; u_5$?

(Bài tập 10/tr105_SGK Đại số 11_Nâng cao)

Đầu tiên nhập dãy số trên vào Maple:

```
> U:={u(n)=u(n-1)-2*u(n-2),u(1)=1,u(2)=-2};
```

```
U:={u(n)=u(n-1)-2*u(n-2),u(1)=1,u(2)=-2}
```

Trước khi tính giá trị các số hạng của dãy trên chúng ta dùng lệnh rsolve để xác định số hạng tổng quát của dãy.

```
> F:=rsolve(U,{u});
```

Bây giờ chúng ta sẽ tính giá trị các số hạng của dãy:

```
> normal(subs(n=3,F));
```

```
{u(3)=-4}
```

```
> normal(subs(n=5,F));
```

```
{u(5)=8}
```

Có thể tính thêm vài số hạng lớn hơn:

```
> normal(subs(n=500,F));
```

```
{u(500)}
```

```
= 4555220279054284213953829836505761769532035716443214  
69645746672394573544
```

```
> normal(subs(n=2007,F));
```

```
{u(2007) =
```

```
-5743708699941719929940313459071184187311336284509256  
045820528554942669315654755482475661118047366445589330  
684722379719033721749851027496175508456562834401988013  
532907765669033223316368823651523554280703381970309049  
322085523593188533482764703112610227889808072359136303  
57133205886536
```

4. Xét tính tăng giảm của một dãy số.

Để xét tính tăng, giảm của một dãy số (u_n) trong Maple, chúng ta xác định dãy như một hàm số để thao tác và biến đổi trên nó dễ dàng hơn.

- Đối với các dãy số cho bởi công thức là đa thức, phân thức không chứa mũ (mũ có n) thì ta kiểm tra dấu của hiệu số $u_{n+1} - u_n$ để xét tính tăng giảm.
- Đối với các dãy có chứa số mũ (số mũ có chứa n) thì chúng ta so sánh tỉ số $\frac{u_{n+1}}{u_n}$ với

1.

Ví dụ 1:

Xét tính tăng, giảm của các dãy số sau:

a) Dãy số (u_n) với $u_n = n^3 - 3n^2 + 5n - 7$;

b) Dãy số (u_n) với $u_n = \frac{n+1}{3^n}$;

c) Dãy số (u_n) với $u_n = \sqrt{n+1} - \sqrt{n}$;

(Bài tập 13/ tr106_SGK ĐS> 11, nâng cao)

• Với dãy số ở câu a)

Định nghĩa dãy số a) trong Maple như một hàm số:

> **u:=n->n^3-3*n^2+5*n-7;**

$$u := n \rightarrow n^3 - 3n^2 + 5n - 7$$

+ Lập hiệu số $u(n+1) - u(n)$:

> **hieu:=u(n+1)-u(n);**

$$hieu := (n+1)^3 - 3(n+1)^2 + 5 - n^3 + 3n^2$$

+ Khai triển và rút gọn:

> **hieu:=simplify(hieu);**

$$hieu := 3n^2 - 3n + 3$$

+ Biến đổi hiệu này về dạng bình phương:

> **hieu:=student[completesquare](hieu,n);**

$$hieu := 3 \left(n - \frac{1}{2} \right)^2 + \frac{9}{4}$$

Kết quả cho ta thấy $u(n+1) - u(n) > 0, \forall n \in \mathbb{N}^*$.

Vậy dãy số đã cho là dãy số tăng.

• Xét dãy số ở câu b)

Định nghĩa dãy số trong Maple:

> **restart;**

> **u:=n->(n+1)/3^n;**

$$u := n \rightarrow \frac{n+1}{3^n}$$

+ Lập tỉ số $\frac{u_{n+1}}{u_n}$ và rút gọn:

> **tiso:=u(n+1)/u(n);**

$$tiso := \frac{(n+2)3^n}{3^{(n+1)}(n+1)}$$

> **tiso:=simplify(%);**

$$tiso := \frac{n+2}{3(n+1)}$$

+ Để so sánh tỉ số trên với 1, ta xét dấu hiệu số $\frac{u_{n+1}}{u_n} - 1$:

```
> `tiso-1`:=simplify(tiso-1);
```

$$tiso-1 := -\frac{2n+1}{3(n+1)}$$

Nhận thấy biểu thức trên nhận giá trị âm với mọi số tự nhiên n .

Suy ra: $\frac{u_{n+1}}{u_n} < 1, \forall n \in \mathbb{N}^*$.

Vậy, dãy số được xét là dãy số giảm.

- Xét dãy số ở câu c)

+ Định nghĩa dãy số trong Maple:

```
> u:=n->sqrtn(n+1)-sqrtn(n);
```

$$u := n \rightarrow \sqrt{n+1} - \sqrt{n}$$

Để xét tính tăng giảm của dãy, chúng ta có thể khảo sát một vài số hạng của dãy:

```
> evalf(seq(u[i]=u(i), i=1..7));
```

$$\begin{aligned} u_1 &= 0.414213562, u_2 = 0.317837246, u_3 = 0.267949192, u_4 \\ &= 0.236067977, u_5 = 0.213421766, u_6 = 0.196261568, u_7 \\ &= 0.182675813 \end{aligned}$$

Qua đó có thể nhận thấy dãy đang xét là dãy số giảm.

Xét hiệu:

```
> u[n+1]-u[n]=u(n+1)-u(n);
```

$$u_{n+1} - u_n = \sqrt{n+2} - 2\sqrt{n+1} + \sqrt{n}$$

Đến đây phải dùng biến đổi đại số để chứng tỏ $u_{n+1} - u_n < 0$ với mọi $n \in \mathbb{N}^*$.

Thật vậy, áp dụng bất đẳng thức Bunhiakopxki, ta có:

$$\sqrt{n+2} + \sqrt{n} < \sqrt{2 \cdot (n+2+n)} = 2\sqrt{n+1} \quad (\text{Dấu "=" không thể xảy ra với } n \in \mathbb{N}^*)$$

Từ đó suy ra $u_{n+1} - u_n < 0$ với mọi $n \in \mathbb{N}^*$. Hay dãy số đã cho giảm.

5. Cấp số cộng, cấp số nhân.

Đối với các dãy số là cấp số cộng hay cấp số nhân. Có thể tính giá trị các dãy đó như dãy bình thường theo một số lệnh như trên.

6. Giới hạn của dãy số.

Sau đây tôi giới thiệu một thủ tục để tìm giới hạn của một dãy số khá đơn giản nhưng hiển thị cả biểu thức lẫn kết quả.

+ Lập thủ tục có tên là 'gh':

```
> gh:=proc(ds,n)
```

```
Limit(ds,n=+infinity)=limit(ds,n=+infinity) end proc;
```

```
gh := proc (ds, n) Limit(ds, n = infinity) = limit(ds, n = infinity) end proc
```

Công dụng, tìm giới hạn của dãy số 'ds' khi $n \rightarrow +\infty$.

Vận dụng thủ tục trên có thể cho hiển thị kết quả của giới hạn $\lim_{n \rightarrow +\infty} \frac{n}{n+1}$ một cách đơn giản

và khá đẹp như sau:

> **gh(n/(n+1),n);**

$$\lim_{n \rightarrow \infty} \frac{n}{n+1} = 1$$

Chỉ cần mất công tạo một thủ tục chúng ta sẽ có một lệnh mới khá gọn gàng và đẹp. Để tự tạo thủ tục quý bạn đọc có thể tham khảo chương cuối (phần phụ lục).

Tính mở của Maple là để cho người dùng có thể tạo ra các hàm, thủ tục riêng. Trên cơ sở đó người dùng có thể tự tạo ra cho mình một số chương trình phục vụ dạy học và nghiên cứu toán.

Một ví dụ khác:

> **gh(sqrt(n^2+1)-sqrt(n^2+n+1),n);**

$$\lim_{n \rightarrow \infty} \sqrt{n^2+1} - \sqrt{n^2+n+1} = \frac{-1}{2}$$